

PRÁCTICA 3:

Cálculo Diferencial de funciones de una variable

En este capítulo vamos a aprender a calcular y evaluar derivadas de cualquier orden de una función; representar gráficamente rectas tangentes y normales a la gráfica de una función; calcular extremos de funciones reales de una variable y, por último, calcular polinomios de Taylor y representarlos gráficamente para aproximar una función.

Cálculo de derivadas

Para calcular la derivada de una función real de variable real, una vez definida, por ejemplo, como $f(x)$, se utiliza el comando `diff` que toma como argumentos la función a derivar, la variable `diff` con respecto a la cual hacerlo y, opcionalmente, el orden de derivación.

<code>diff(expr,variable)</code>	derivada de <i>expr</i>
<code>diff(expr,variable,n)</code>	derivada <i>n</i> -ésima de <i>expr</i>

A este comando también podemos acceder a través del menú **Análisis→Derivar** o, también, a través del botón **Derivar** si hemos activado el correspondiente panel. Haciéndolo de cualquiera de estas dos formas, aparece una ventana de diálogo con varios datos a rellenar; a saber:

- Expresión. Por defecto, `wxMaxima` rellena este espacio con `%` para referirse a la salida anterior. Si no es la que nos interesa, la escribimos directamente nosotros.
- respecto la variable. Se refiere a la variable respecto a la cual vamos a derivar.
- veces. Se refiere al orden de derivación.

Comencemos con un ejemplo,




Figura Introducir derivada

```
(%i1) diff(tan(x),x);
(%o1) sec(x)^2
```

La orden `diff` considera como constantes cualquier otra variable que aparezca en la expresión a derivar, salvo que explícitamente manifestemos que están relacionadas.

```
(%i2) diff(a*x*sin(a+x),x);
(%o2) a sin(x+a)+a x cos(x+a)
```

También podemos trabajar con funciones que previamente hayamos definido.

```
(%i3) f(x):= x^4+sin(x^2);
(%o3) f(x):=x^4+sin(x^2)
(%i4) diff(f(x),x);
(%o4) 2x cos(x^2)+4x^3
```

Práctica 3: Cálculo Diferencial de funciones de una variable

La tercera entrada de la orden `diff` nos permite calcular derivadas de orden superior. Por ejemplo, la cuarta derivada de f sería la siguiente.

```
(%i5) diff(f(x),x,4);  
(%o5) 16x^4sin(x^2) - 12sin(x^2) - 48x^2cos(x^2) + 24
```

Reutilizar la derivada

Las derivadas sucesivas de una función nos dan mucha información sobre la función original y con frecuencia nos hace falta utilizarlas de nuevo, ya sea para calcular puntos críticos, evaluar para estudiar monotonía o extremos relativos, etc. Es por ello que es cómodo escribir la derivada como una función. Hay varias formas en las que podemos hacerlo. Nosotros vamos a utilizar la orden `define`

```
(%i6) define(g(x),diff(f(x),x));  
(%o6) g(x):=2xcos(x^2)+4x^3
```

Recta tangente y normal a una función

Ejemplo. Consideremos la función $f(x) = x^3 - 2x^2 - x + 2$ y su derivada, a la que notaremos df ,


```
(%i19) f(x):=x^3-2*x^2-x+2;  
(%o19) f(x):=x^3-2x^2-x+2  
(%i20) define(df(x),diff(f(x),x));  
(%o20) df(x):=3x^2-4x-1
```

La recta tangente a una función f en un punto a es la recta $y = f(a) + f'(a)(x - a)$. La definimos.

```
(%i21) tangente(x,a):=f(a)+df(a)*(x-a);  
(%o21) tangente(x,a):=f(a)+df(a)(x-a)
```


Ya podemos dibujar la función y su tangente en 1:

```
(%i12) wxplot2d([f(x),tangente(x,1)], [x,-2,3]);
```



La recta normal es la recta perpendicular a la recta tangente. Su pendiente es $-1/f'(a)$ y, por tanto, tiene como ecuación $y = f(a) - \frac{1}{f'(a)}(x - a)$. Con todo lo que ya tenemos hecho es muy fácil dibujarla.

```
(%i14) normal(x,a):=f(a)-df(a)^(-1)*(x-a)$
(%i17) wxplot2d([f(x),normal(x,1)], [x,-2,3]);
```



En la figura anterior, la recta normal no parece perpendicular a la recta tangente. Eso es por que no hemos tenido en cuenta la escala a la que se dibujan los ejes. Prueba a cambiar la escala para que queden perpendiculares.


Máximos y mínimos relativos

En esta sección vamos a aprender a localizar extremos relativos de una función f . Para ello encontraremos las soluciones de la ecuación de punto crítico: $f'(x) = 0$. Y para resolver dicha ecuación podemos usar el comando `solve`.

Ejemplo Calculemos los extremos relativos de la función $f(x) = 2x^4 + 4x$, $\forall x \in \mathbb{R}$.

Comenzamos, entonces, presentándosela al programa (no olvidéis borrar de la memoria la anterior función f) y pintando su gráfica para hacernos una idea de dónde pueden estar sus extremos.

```
(%i29) f(x):=2*x^4+4*x$
(%i30) plot2d(f(x), [x,-2,1]);
(%o30)
```



Parece que hay un mínimo en las proximidades de -1. Para confirmarlo, calculamos los puntos críticos de f .

```
(%i31) define(d1f(x),diff(f(x),x))$
(%i32) puntosf:solve(d1f(x),x);
(%o32) [x=(-sqrt(3)*%i-1)/(2*2^(1/3)),x=(sqrt(3)*%i+1)/(2*2^(1/3)),
x=-1/2^(1/3)]
```

Observamos que hay sólo una raíz real que es la única que nos interesa.

```
(%i11) puntosf[3];
(%o11) x = -1/2^(1/3)
```

Práctica 3: Cálculo Diferencial de funciones de una variable

Llamamos a al único punto crítico que hemos obtenido, y evaluamos en él la segunda derivada: Para seleccionar la parte derecha de la ecuación anterior es muy cómodo usar el comando

```
rhs (expresión1=expresión2)  expresión2
```

```
(%i34)  a:rhs(%)$
(%i35)  define(d2f(x),diff(f(x),x,2))$
(%i36)  d2f(a);
(%o36)  24/2^(2/3)
```


Por tanto la función f tiene un mínimo relativo en dicho punto a . ¿Puede haber otro extremo más? ¿Cómo podemos asegurarnos? <

Ejemplo Vamos a calcular los extremos relativos de la función:

$$f(x) = x^5 + x^4 - 11x^3 - 9x^2 + 18x + 10 \text{ en el intervalo } [-4, 4]$$

Procedemos de la misma forma que en el ejemplo anterior.

```
(%i37)  f(x):=x^5+x^4-11*x^3-9*x^2+18*x+10$
(%i38)  plot2d(f(x),[x,-4,4])$
(%o38)
```



A simple vista observamos que hay:

a) un máximo relativo entre -3 y -2, b) un mínimo relativo entre -2 y -1, c) un máximo relativo entre 0 y 1, y d) un mínimo relativo entre 2 y 3.

Vamos entonces a derivar la función e intentamos calcular los ceros de f' haciendo uso del comando `solve`.

```
(%i39)  define(d1f(x),diff(f(x),x));
(%o39)  d1f(x):=5x^4+4x^3-33x^2-18x+18
(%i40)  solve(d1f(x),x);
```

Las soluciones que nos da el programa no son nada manejables, así que vamos a intentar resolver la ecuación $f'(x)=0$ de forma numérica, en primer lugar con la ayuda del comando **algsys**.

```
(%i3)  raices:algsys([d1f(x)], [x]);
(%o3)  [[x=-2.600820933812211],[x=2.363816114681167],[x=-1.096856414613424],
, [x=0.53386138613861]]
```

Como vemos, en este ejemplo, el comando **algsys** sí es capaz de encontrar los puntos críticos. Ahora utilizamos el criterio de la derivada segunda para ver si estos puntos críticos son máximos o mínimos relativos.

Práctica 3: Cálculo Diferencial de funciones de una variable

```
(%i4) define(d2f(x),diff(f(x),x,2));
(%o4) d2f(x):=20 x^3+12 x^2-66 x-18

(%i5) d2f(x),raices[1];
(%o5) -117.0276599100646

(%i6) d2f(x),raices[2];
(%o6) 157.2020930532604

(%i7) d2f(x),raices[3];
(%o7) 42.43722399568166

(%i8) d2f(x),raices[4];
(%o8) -46.77166062872889
```

Por tanto en el primer y cuarto puntos hay máximo relativo, mientras que en los otros dos tenemos mínimos relativos.

Nota: En otros ejercicios es posible que el comando **algsys** tampoco sea capaz de localizar los ceros de la derivada primera. En estos ejemplos será necesario recurrir al comando **find_root**. Si lo aplicamos al ejemplo anterior la idea es apoyarse en la gráfica anterior buscando intervalos en los que es posible encontrar solución. En este caso se obtiene:

```
(%i41) puntosf:[find_root(d1f(x),x,-3,-2),find_root(d1f(x),x,-2,1),
                find_root(d1f(x),x,0,1),find_root(d1f(x),x,2,3)];
(%o41) [-2.600821117505113,-1.096856508567827,0.53386135374308,
        2.363816272329865]
```

Polinomio de Taylor

Polinomio de Taylor de orden n de una función f en un punto a :

$$\begin{aligned} T(f,a,n)(x) &= f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 \\ &+ \frac{f'''(a)}{3!}(x-a)^3 + \dots + \frac{f^{(n)}(a)}{n!}(x-a)^n \\ &= \sum_{k=1}^n \frac{f^{(k)}(a)}{k!}(x-a)^k \end{aligned}$$

El programa tiene una orden que permite calcular directamente el polinomio de Taylor centrado en un punto a . Se trata del comando **taylor**. En concreto, el comando **taylor(f(x),x,a,n)** nos da el polinomio de Taylor de la función f centrado en a y de grado n . Haciendo uso del menú podemos acceder al comando anterior desde **Análisis—Calcular serie**. Entonces se abre una ventana de diálogo en la que, escribiendo la expresión de la función, la variable, el punto en el que desarrollamos y el orden del polinomio de Taylor, obtenemos dicho polinomio. Como en otras ventanas similares, si marcamos la casilla de **Especial**, podemos elegir π o e como centro para el cálculo del desarrollo.




Figura Ventana para el cálculo del polinomio de Taylor

taylor(f(x),x,a,n)	polinomio de Taylor de la función f en el punto a y de orden n
trunc(polimonio)	convierte polinomio de Taylor en un polinomio

Práctica 3: Cálculo Diferencial de funciones de una variable


Veamos un ejemplo.

```
(%i67)  taylor(cos(x),x,0,5);
```

```
(%o67)   $1 - \frac{x^2}{2} + \frac{x^4}{24} + \dots$ 
```


En teoría, un polinomio de Taylor de orden más alto debería aproximar mejor a la función. Ya hemos visto cómo aproxima la recta tangente a la función coseno. Vamos ahora a dibujar las gráficas de la función $f(x) = \cos(x)$ y de su polinomio de Taylor de orden 8 en el cero para comprobar que la aproximación es más exacta.

```
(%i69)  plot2d([f(x),taylor(f(x),x,0,8)], [x,-4,4], [y,-2,2]);
```




Pero si aumentamos el dominio podemos ver que el polinomio de Taylor se separa de la función cuando nos alejamos del origen.

```
(%i70)  plot2d([f(x),taylor(f(x),x,0,8)], [x,-8,8], [y,-2,2]);
```



Esto es lo esperable: la función coseno está acotada y el polinomio de Taylor, como todo polinomio no constante, no lo está. Eso sí, si aumentamos el grado del polinomio de Taylor vuelven a parecerse:

```
(%i71)  plot2d([f(x),taylor(f(x),x,0,14)], [x,-8,8], [y,-2,2]);
```




Práctica 3: Cálculo Diferencial de funciones de una variable

Observación *Maxima* tiene dos formas de representar internamente los polinomios. Sin entrar en detalles, no se guardan de la misma forma un polinomio de Taylor y un polinomio cualquiera. Esto puede dar lugar a algunas sorpresas. Por ejemplo, hemos visto cómo el polinomio de Taylor nos sirve para aproximar una función, pero, en lugar de representar la función y dicho polinomio, podríamos representar la diferencia. Veamos que ocurre⁹. Definimos las funciones,

```
(%i74) f(x):=cos(x);  
(%o74) f(x):=cos(x)  
(%i75) define(g(x),taylor(f(x),x,0,5));  
(%o75) g(x):=1- $\frac{x^2}{2}+\frac{x^4}{24}+\dots$ 
```

y dibujamos la diferencia

```
(%i76) plot2d(f(x)-g(x),[x,-5,5]);
```




¿Cómo puede salir 0? ¿Es que no hay diferencia? Si la hay. Ya lo sabemos: si evaluamos en algún punto podemos ver que el resultado no es cero.

```
(%i77) f(2)-g(2);  
(%o77)  $\frac{3 \cos(2) + 1}{3}$ 
```

Como hemos avisado antes, *Maxima* maneja de forma diferente un polinomio de Taylor y un polinomio “normal”.

trunc La orden `trunc(polinomio de Taylor)` nos permite pasar de polinomio de Taylor a polinomio “normal”. Con estos ya no tenemos este problema.

```
(%i80) plot2d(f(x)-trunc(g(x)),[x,-5,5]);  
(%o80)
```



EJERCICIOS PROPUESTOS

1) Hallar las derivadas por la derecha y por la izquierda de la función $f(x) = |x - 1|$ en el punto $x = 1$. Representarla gráficamente en el intervalo $[0, 2]$.

2) Estudiar si la siguiente función es derivable en los puntos $x = -1, x = 1, x = 0$.

$$f(x) = \begin{cases} 1 & \text{si } x < -1 \\ |x| & \text{si } -1 \leq x \leq 1 \\ -1 & \text{si } 1 \leq x \end{cases}$$

Representar la gráfica en el intervalo $[-3, 3]$.

3) Dada la función:

$$f(x) = x^2 \operatorname{sen}\left(\frac{1}{x^2}\right) + 3 \cos x + x \quad \text{si } x \neq 0 \quad \text{y} \quad f(0) = 3$$

a) Comprobar la continuidad en $x = 0$.

b) Hallar la derivada en $x = 3$, siendo ésta el límite:

$$\lim_{x \rightarrow 3} \frac{f(x) - f(3)}{x - 3}$$

c) Calcular el corte de la gráfica con el eje de abscisas (con *find_root*) y dibujar la gráfica de f junto con la tangente en ese punto.

4) Hallar máximos, mínimo y puntos de inflexión de la función:

$$f(x) = x^3 - x^5$$

Comprobarlo mediante la representación gráfica.

5) Hallar los extremos relativos y los puntos de inflexión de la función:

$$f(x) = 10x^6 - 24x^5 + 15x^4 + 2$$

6) Dibujar en una misma ventana la función:

$$f(x) = x \cos(3x)$$

y sus polinomios de McLaurin de orden 2, 4, 6 y todos con diferentes colores.